
Module 19.2: nag short path

Shortest Path Problems

nag short path contains a procedure for finding the shortest path through a directed

or undirected acyclic network.

Introduction

This module is concerned with finding the shortest path through a directed or undirected acyclic network,
which consists of a set of points called vertices and a set of arcs that connect certain pairs of distinct
vertices. An acyclic network is one in which there are no paths connecting a vertex to itself. An arc
whose origin vertex is i and whose destination vertex is j can be written as i → j. In an undirected
network the arcs i → j and j → i are equivalent (i.e., i ↔ j), whereas in a directed network they are
different. Note that the shortest path may not be unique and in some cases may not even exist (e.g., if
the network is disconnected).

Procedure: nag short path find

1 Description

nag short path find attempts to determine a path ns → ne between two specified vertices ns and ne

of shortest length.

The network is assumed to consist of n vertices which are labelled by the integers 1, 2, . . . , n. The lengths
of the arcs between the vertices are defined by the n by n distance matrix D, in which the element dij

gives the length of the arc i → j; dij = 0 if there is no arc connecting vertices i and j (as is the case for
an acyclic network when i = j). For example, if n = 4 and the network is directed, then

D =









0 d12 d13 d14

d21 0 d23 d24

d31 d32 0 d34

d41 d42 d43 0









.

If the network is undirected, D is symmetric since dij = dji (i.e., the length of the arc i → j ≡ the length
of the arc j → i).

As the matrix D is usually sparse, only the non-zero elements are required and must be supplied via the
arguments distance, row index and col index.

The method used by nag short path find is described in detail in Section 6.1.

Let m denote the number of vertices between ns and ne for which the path ns → ne is of shortest
length. The argument path is a pointer array because the exact value of m is not known in
advance. The procedure allocates the required amount of memory to path; on exit from the procedure,
m = SIZE(path).

2 Usage

USE nag short path

CALL nag short path find(num vertex, distance, row index, col index, &

short path len, path [, optional arguments])

3 Arguments

Note. All array arguments are assumed-shape arrays. The extent in each dimension must be exactly that required by
the problem. Notation such as ‘x(n)’ is used in the argument descriptions to specify that the array x must have exactly n

elements.

This procedure derives the values of the following problem parameters from the shape of the supplied
arrays.

nz — the number of non-zeros

nz must satisfy the constraint 1 ≤ nz ≤ pn(n− 1), where p = 1 if the network is directed and p = 1/2 if
the network is undirected (see the optional argument undirected).

3.1 Mandatory Arguments

num vertex — integer, intent(in)

Input: n, the number of vertices.

Constraints: num vertex ≥ 2.

distance(nz) — real(kind=wp), intent(in)

Input: the non-zero elements of the distance matrix D. More precisely, distance(k) must contain
the value of the non-zero element with indices (row index(k),col index(k)); this is the length of
the arc from the vertex with label row index(k) to the vertex with label col index(k). Elements
with the same row and column indices are not allowed. If undirected = .true.(see Section 3.2),
then only those non-zero elements in the strict upper triangle of D need be supplied since dij = dji.

Constraints: distance(k) > 0.0, for k = 1, 2, . . . , nz.

row index(nz) — integer, intent(in)

col index(nz) — integer, intent(in)

Input: row index(k) and col index(k) must contain the row and column indices, respectively, for
the non-zero element stored in distance(k).

Constraints:

if undirected = .false., 1 ≤ row index(k) ≤ num vertex, 1 ≤ col index(k) ≤ num vertex

and row index(k) 6= col index(k);

if undirected = .true., 1 ≤ row index(k) < col index(k) ≤ num vertex.

short path len — real(kind=wp), intent(out)

Output: the length of the shortest path between the specified vertices ns and ne.

path(:) — integer, pointer

Output: contains details of the shortest path between the specified vertices ns and ne. More
precisely, first vertex = path(1)→ path(2)→ . . . → path(m) = last vertex.

Note: the procedure creates a target array of shape (m).

3.2 Optional Arguments

Note. Optional arguments must be supplied by keyword, not by position. The order in which they are described below
may differ from the order in which they occur in the argument list.

first vertex — integer, intent(in), optional

last vertex — integer, intent(in), optional

Input: ns and ne, the labels of the first and last vertices respectively, between which the shortest
path is sought.

Constraints:

1 ≤ first vertex ≤ num vertex,

1 ≤ last vertex ≤ num vertex,

first vertex 6= last vertex.

Default: first vertex = 1 and last vertex = num vertex.

undirected — logical, intent(in), optional

Input: specifies whether the network is directed or undirected as follows:

if undirected = .true., then the network is undirected;

if undirected = .false., then the network is directed.

Default: undirected = .false..

error — type(nag error), intent(inout), optional

The NAG fl90 error-handling argument. See the Essential Introduction, or the module document
nag error handling (1.2). You are recommended to omit this argument if you are unsure how to
use it. If this argument is supplied, it must be initialized by a call to nag set error before this
procedure is called.

4 Error Codes

Fatal errors (error%level = 3):

error%code Description

301 An input argument has an invalid value.

302 An array argument has an invalid shape.

303 Array arguments have inconsistent shapes.

320 The procedure was unable to allocate enough memory.

Failures (error%level = 2):

error%code Description

201 No connected network exists between vertices first vertex and last vertex.

5 Examples of Usage

A complete example of the use of this procedure appears in Example 1 of this module document.

6 Further Comments

6.1 Algorithmic Detail

This procedure is based upon Dijkstra’s algorithm (see [1]), which attempts to find a path ns → ne

between two specified vertices ns and ne of shortest length d(ns, ne).

The algorithm proceeds by assigning labels to each vertex, which may be temporary or permanent . A
temporary label can be changed, whereas a permanent one cannot. For example, if vertex p has a
permanent label (q, r) then r is the distance d(ns, r) and q is the previous vertex on a shortest length
ns → p path. If the label is temporary, then it has the same meaning but it refers only to the shortest
ns → p path found so far. A shorter one may be found later, in which case the label may become
permanent.

The algorithm consists of the following steps.

1. Assign the permanent label (−, 0) to vertex ns and temporary labels (−,∞) to every other vertex.
Set k = ns and go to 2.

2. Consider each vertex y adjacent to vertex k with a temporary label in turn. Let the label at k be
(p, q) and at y (r, s). If q + dky < s, then a new temporary label (k, q + dky) is assigned to vertex y;
otherwise no change is made in the label of y. When all vertices y with temporary labels adjacent to k
have been considered, go to 3.

3. From the set of temporary labels, select the one with the smallest second component and declare
that label to be permanent. The vertex it is attached to becomes the new vertex k. If k = ne go to 4.
Otherwise go to 2 unless no new vertex can be found (e.g., when the set of temporary labels is ‘empty’
but k 6= ne, in which case no connected network exists between vertices ns and ne).

4. To find the shortest path, let (y, z) denote the label of vertex ne. The column label (z) gives d(ns, ne)
while the row label (y) then links back to the previous vertex on a shortest length ns → ne path. Go to
vertex y. Suppose that the (permanent) label of vertex y is (w, x), then the next previous vertex is w
on a shortest length ns → y path. This process continues until vertex ns is reached. Hence the shortest
path is

ns → . . . → w → y → ne,

which has length d(ns, ne).

6.2 Accuracy

The results are exact, except for the obvious errors in summing the distances in the length of the shortest
path.

Example 1: Shortest Path Problem

To find the shortest path between vertices 1 and 11 for the undirected network

1 Program Text

Note. The listing of the example program presented below is double precision. Single precision users are referred to
Section 5.2 of the Essential Introduction for further information.

PROGRAM nag_short_path_ex01

! Example Program Text for nag_short_path

! NAG fl90, Release 4. NAG Copyright 2000.

! .. Use Statements ..

USE nag_examples_io, ONLY : nag_std_in, nag_std_out

USE nag_short_path, ONLY : nag_short_path_find

! .. Implicit None Statement ..

IMPLICIT NONE

! .. Intrinsic Functions ..

INTRINSIC KIND

! .. Parameters ..

INTEGER, PARAMETER :: wp = KIND(1.0D0)

! .. Local Scalars ..

INTEGER :: j, nnz, num_vertex

REAL (wp) :: short_path_len

LOGICAL :: undirected

! .. Local Arrays ..

INTEGER, ALLOCATABLE :: col_index(:), row_index(:)

INTEGER, POINTER :: path(:)

REAL (wp), ALLOCATABLE :: distance(:)

! .. Executable Statements ..

WRITE (nag_std_out,*) ’Example Program Results for nag_short_path_ex01’

READ (nag_std_in,*) ! Skip heading in data file

! Read number of non-zeros (nnz)

READ (nag_std_in,*) nnz

ALLOCATE (col_index(nnz),row_index(nnz),distance(nnz)) ! Allocate storage

! Read in problem data

READ (nag_std_in,*) num_vertex, undirected

READ (nag_std_in,*) (distance(j),row_index(j),col_index(j),j=1,nnz)

! Find the shortest path between vertices 1 and num_vertex

CALL nag_short_path_find(num_vertex,distance,row_index,col_index, &

short_path_len,path,undirected=undirected)

! Print details of shortest path

WRITE (nag_std_out,fmt=’(/ 1x, a, 10(i2, :, ’’ to ’’))’) &

’Shortest path = ’, path

WRITE (nag_std_out,fmt=’(/ 1x, a, g16.7 )’) ’Length of shortest path = ’ &

, short_path_len

DEALLOCATE (col_index,row_index,distance,path) ! Deallocate storage

END PROGRAM nag_short_path_ex01

2 Program Data

Example Program Data for nag_short_path_ex01

20 : nnz

11 .TRUE. : num_vertex, undirected

6.0 6 8

1.0 8 9

2.0 9 11

4.0 2 5

1.0 3 4

6.0 1 3

4.0 3 6

1.0 4 6

2.0 2 3

3.0 4 7

5.0 1 2

7.0 6 10

1.0 5 6

4.0 8 11

9.0 5 9

1.0 6 7

8.0 7 9

4.0 10 11

2.0 9 10

5.0 1 4 : distance, row_index, col_index

3 Program Results

Example Program Results for nag_short_path_ex01

Shortest path = 1 to 4 to 6 to 8 to 9 to 11

Length of shortest path = 15.00000

